Example 8.9 Heat Conduction with the Time Dependent Boundary Conditions
Equation (8.1.19) is solved by slightly modifying the Maple program used for example 8.7 as:

```
> restart : with(inttrans) : with(plots) :
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t} u(x, t) = \frac{\partial^2}{\partial x^2} u(x, t) \tag{1}$$

```
> u(x,0):=0;
```

$$u(x, 0) := 0 \tag{2}$$

```
> bc1:=diff(u(x,t),x)=0;
```

$$bc1 := \frac{\partial}{\partial x} u(x, t) = 0 \tag{3}$$

```
> bc2:=u(x,t)=exp(-t);
```

$$bc2 := u(x, t) = e^{-t} \tag{4}$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\, U(x) = \frac{d^2}{dx^2} U(x) \tag{5}$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := \frac{d}{dx} U(x) = 0 \tag{6}$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := U(x) = \frac{1}{1+s} \tag{7}$$

```
> dsolve(eqs,U(x));
```

$$U(x) = \_C1\, e^{\sqrt{s}\, x} + \_C2\, e^{-\sqrt{s}\, x} \tag{8}$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x);
```

$$U(x) := c_1 \cosh\left(\sqrt{s}\, x\right) + c_2 \sinh\left(\sqrt{s}\, x\right) \tag{9}$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := \frac{\cosh\left(\sqrt{s}\, x\right)}{\cosh\left(\sqrt{s}\right)(1+s)} \tag{10}$$

```
> P(s):=numer(U(x));
```

$$P(s) := \cosh\left(\sqrt{s}\, x\right) \tag{11}$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := \cosh\left(\sqrt{s}\right)\,(1+s) \tag{12}$$

```
> A(s):=P(s)/diff(Q(s),s);
```

$$A(s) := \frac{\cosh\left(\sqrt{s}\,x\right)}{\frac{1}{2}\,\frac{\sinh\left(\sqrt{s}\right)\,(1+s)}{\sqrt{s}} + \cosh\left(\sqrt{s}\right)} \tag{13}$$

```
> solve(Q(s),s);
```

$$-\frac{1}{4}\,\pi^2,\ -1 \tag{14}$$

```
> _EnvAllSolutions := true:
> solve(Q(s),s):
```

The roots are:

```
> -1,-((2*n-1)*Pi/2)^2;
```

$$-1,\ -\frac{1}{4}\,(2\,n-1)^2\,\pi^2 \tag{15}$$

```
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2\cosh\left(\sqrt{\mu}\,x\right)\sqrt{\mu}}{\sinh\left(\sqrt{\mu}\right) + \sinh\left(\sqrt{\mu}\right)\mu + 2\cosh\left(\sqrt{\mu}\right)\sqrt{\mu}} \tag{16}$$

```
> A[0]:=subs(mu^(1/2)=I,mu=-1,A[n]):
> A[0]:=simplify(A[0]);
```

$$A_0 := \frac{\cos(x)}{\cos(1)} \tag{17}$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*(2*n-1)/2*Pi,mu=-((2*n-1)*Pi/2)
  ^2,A[n])):
> vars:={cos(1/2*(2*n-1)*Pi)=0,sin(1/2*(2*n-1)*Pi)=(-1)^(n-1)}:
> A[n]:=simplify(subs(vars,A[n]));
```

$$A_n := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)}{-4+4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2} \tag{18}$$

```
> u0s:=A[0]*subs(mu=-1,1/(s-mu));
```

$$u0s := \frac{\cos(x)}{\cos(1)\,(1+s)} \tag{19}$$

```
> u0t:=invlaplace(u0s,s,t);
```

$$u0t := \frac{\cos(x)\,\mathrm{e}^{-t}}{\cos(1)} \tag{20}$$

```
> uns:=A[n]/(s-mu);
```

$$uns := \frac{4\,(-1)^{-n}\,(2\,n-1)\,\pi\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)}{\left(-4+4\,\pi^2\,n^2-4\,\pi^2\,n+\pi^2\right)\,(s-\mu)} \qquad \textbf{(21)}$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{4\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)\,e^{\mu\,t}\,\pi\,(-1)^{-n}\,(2\,n-1)}{(2\,\pi\,n-\pi)^2-4} \qquad \textbf{(22)}$$

```
> unt:=subs(mu=-((2*n-1)/2*Pi)^2,unt);
```

$$unt := \frac{4\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)\,e^{-\frac{1}{4}\,(2\,n-1)^2\,\pi^2\,t}\,\pi\,(-1)^{-n}\,(2\,n-1)}{(2\,\pi\,n-\pi)^2-4} \qquad \textbf{(23)}$$

```
> U:=u0t+Sum(unt,n=1..infinity);
```

$$U := \frac{\cos(x)\,e^{-t}}{\cos(1)}+\sum_{n=1}^{\infty}\frac{4\cos\left(\frac{1}{2}\,(2\,n-1)\,\pi\,x\right)\,e^{-\frac{1}{4}\,(2\,n-1)^2\,\pi^2\,t}\,\pi\,(-1)^{-n}\,(2\,n-1)}{(2\,\pi\,n-\pi)^2-4} \qquad \textbf{(24)}$$

```
> u:=piecewise(t=0,0,t>0,subs(infinity=20,U)):
```

The following plots are obtained:

```
> plot3d(u,x=0..1,t=0..1,axes=boxed,title="Figure 8.16.",labels=
  [x,t,"u"],orientation=[-150,50]);
```

Figure 8.16.



The dimensionless temperature at the surface x =0 reaches a maximum and then decreases as a function of time:

```
> plot(subs(x=0,u),t=0..2,thickness=3,axes=boxed,title="Figure
  Exp. 8.17.",labels=[t,"u(0,t)"]);
```